CS 61A MIDTERM 2 REVIEW SESSION (SOLUTIONS) July 26, 2006

Exercise 0. What is your TA's name (spelled correctly!)? What is your discussion section number?

Gap Thirathon, Ramesh Sridharan

Exercise 1. What will Scheme print in response to the following expressions? If an expression produces an error message, you may just write "error"; you don't have to provide the exact text of the message. If the value of an expression is a procedure, just write "procedure"; you don't have to show the form in which Scheme prints procedures. **If the return value is a pair, then draw the corresponding box and pointer diagram.**

```
a. ((caddr '(1 + * 2)) 3 4)
; ERROR
; The "*" that is returned by the caddr is the symbol "*" and not the procedure "*".
; This would have worked if we had entered (list 1 + * 2) instead of '(1 + * 2). (Why?)

b. (list? (cons 1 (cons 2 (cons nil nil))))
#t

c. (list (list (list (cons 1 nil))))
((((1))))

d. (list 'one (cons 'foo (list '())))
(one (foo ()))

e. (filter number? '(1 (a 2) ((b) 3) c 4 d))
(1 4)

f. '(a (b c d) ((e . f) g))
(a (b c d) ((e . f) g))

g. (cddadr '(a (((b)) c d e f) g))
(d e f)

h. (cdaadr '(a ((b . c)) (d)))
c

i. (if (= 2 3) '(1 2) '(3 4))
(3 4)

j. (map (lambda (x) (cons 'foo x)) '(1 (2) (3)))
((foo . 1) (foo 2) (foo 3))
```

Exercise 2. Write a predicate procedure `deep-car?` that takes a symbol and a deep-list (possibly including sublists to any depth) as its arguments. It should return true if and only if the symbol is the `car` of the list or of some list that's an element, or an element of an element, etc.

```
> (deep-car? 'a '(a b c))
#t
> (deep-car? 'a (b (c a) a d))
#f
> (deep-car? 'a ((x y) (z (a b) c) d))
#t
```

Fill in the blanks to complete the definition.

```
(define (deep-car? symbol lst)
  (if (pair? lst)
      (or (eq? symbol (CAR LST))
          (helper symbol LST))
      #F))

(define (helper symbol lsts)
  (cond ((null? lsts) #F)
        ((deep-car? symbol (car lsts))
         #T)
        (else (HELPER SYMBOL (CDR LSTS)))))
```

Exercise 3. An **athlete** contains 3 pieces of information: **name** (a word), **country** (a word), and **events** (a sentence containing the names of all the events the athlete participates in). The constructor:

```
(define (make-athlete name country events)
  (append (list name country) (cons events '())))
```

(a) Write the appropriate selectors for the athlete abstract data type.

```
(define (name athlete) (car athlete))
(define (country athlete) (cadr athlete))
(define (events athlete) (caddr athlete))
```

(b) Write a procedure `(get-participants event athletes)` that returns a sentence containing the names of all the athletes which participate in that event.

```
(define (get-participants event list-of-athletes)
  (cond ((null? list-of-athletes) '())
        ((member? event (events (car list-of-athletes)))
         (sentence (name (car list-of-athletes))
                   (get-participants event (cdr list-of-athletes))))
        (else (get-participants event (cdr list-of-athletes)))))

; Note that we use SENTENCE here instead of CONS.
; This is because we want to return a sentence, not a list.
; Using CONS would be a data abstraction violation.
```

Exercise 4. Here is the code for `eval-1`.

```
(define (eval-1 exp)
  (cond ((constant? exp) exp)
        ((symbol? exp) (eval exp))
        ((quote-exp? exp) (cadr exp))
        ((if-exp? exp)
           (if (eval-1 (cadr exp))
               (eval-1 (caddr exp))
               (eval-1 (cadddr exp))))
        ((lambda-exp? exp) exp)
        ((pair? exp)
           (apply-1 (eval-1 (car exp))
                    (map eval-1 (cdr exp))))
        (else (error "bad expr: " exp))))
```

Suppose we moved the moved the `cond` clause beginning with `pair?` (the one that handles function calls) to the beginning of the `cond`, making it the first clause:

```
(define (eval-1 exp)
  (cond ((pair? exp) ...)
        ((constant? exp) ...)
        ((symbol? exp) ...)
        ...
        (else (error ...))))
```

With the above changes, show what Scheme-1 would print given the following inputs. If the result is an error, just write "error".

```
Scheme-1: 3
3

Scheme-1: +
#[closure ...] ; The value returned is an STk procedure, not a plus sign!

Scheme-1: (if #t 1 2)
ERROR

Scheme-1: ((lambda (x) (* x x)) 3)
ERROR

Scheme-1: (+ 3 4)
7

Scheme-1: (lambda (x) (* x x))
ERROR
```

Exercise 5. Write the function `datum-filter` which, given a predicate and a tree, returns a list of all the data (that's plural for datum!) which satisfy the predicate (in any order). Do this for general trees, not binary trees. The function should return the empty list for any tree in which no data satisfy the predicate. You may use helper procedures.

```
(define (datum-filter pred tree)
  (if (pred (datum tree))
      (cons (datum tree) (forest-datum-filter pred (children tree)))
      (forest-datum-filter pred (children tree))))

(define (forest-datum-filter pred forest)
  (if (null? forest)
      '()
      (append (datum-filter pred (car forest)) (forest-datum-filter pred (cdr forest)))))
```

Exercise 6. Write `double-datum?`, which takes a tree as its argument, and returns true if there exists a node anywhere in the tree that has the same datum as one of its children (immediate children, not grandchildren or cousins), otherwise false.

```
(define (double-datum? tree)
  (or (member (datum tree) (map datum (children tree)))
      (double-datum-forest? (children tree))))
```

```
(define (double-datum-forest? forest)
  (cond ((null? forest) #f)
        ((double-datum? (car forest)) #t)
        (else (double-datum-forest? (cdr forest)))))
```

Exercise 7. Instead of attaching one tag to an object, suppose we wanted to attach multiple tags. After all, a tomato is not only a food, it's also a projectile! So we'll rename `attach-tag` to `attach-tags` (taking a list of tags as its first argument), and rename `type-tag` to `type-tags`.

Here is the original version of `operate`:

```
(define (operate op obj)
  (let ((proc (get (type-tag obj) op)))
    (if proc
        (proc (contents obj))
        (error "No such operator for this type"))))
```

Rewrite `operate` so that it looks for each of an object's type tags in the table, using the first one for which a procedure is found for this operator.

```
(define (operate op obj)
  (define (multi-get tags)
    (cond ((null? tags) (error "No such operator for these types"))
          ((get (car tags) op) ((get (car tags) op) (contents obj)))
          (else (multi-get (cdr tags)))))
  (multi-get (type-tags obj)))
```

Exercise 8. Write a procedure `(flatten ls)` that takes in a deep list and flattens all of the elements into a flat list.

```
(define (flatten ls)
  (cond ((null? ls) '())
        ((pair? (car ls)) (append (flatten (car ls)) (flatten (cdr ls))))
        (else (cons (car ls) (flatten (cdr ls))))))
```

Exercise 9. Below are definitions of `m-car` and `m-cdr` (the `m` here stands for "message-passing").

```
(define (m-car pair) (pair 'car))
(define (m-cdr pair) (pair 'cdr))
```

(a) Write a definition of `m-cons` that behaves just like regular `cons` when used in conjunction with `m-car` and `m-cdr`.

```
(define (m-cons car cdr)
  (lambda (msg) (if (equal? msg 'car) car cdr)))
```

(b) If you type `list1` at the STk prompt, what would it print out?

```
> (define list1 (m-cons 1 (m-cons 2 (m-cons 3 '()))))
#[closure ...]
```

(c) We would like to be able to convert a message-passing list into an ordinary Scheme list. Write the function `m-list-to-regular-list` that takes a message-passing list (hmm, I haven't yet defined what a message-passing list is; how should it be defined?) and returns a regular list of the same elements.

```
(define (m-list-to-regular-list m-list)
  (if (null? m-list)
      '()
      (cons (m-list 'car) (m-list-to-regular-list (m-list cdr)))))

; This only works for flat m-lists; how can we generalize it to DEEP m-lists?
; Is it even possible? If so, to what extent is it possible?
```

Exercise 10. Here is the code for `apply-1`.

```
(define (apply-1 proc args)
  (cond ((procedure? proc)
         (apply proc args))
        ((lambda-exp? proc)
         (eval-1 (substitute (caddr proc)  ; the body
                             (cadr proc)   ; the formal parameters
                             args          ; the actual arguments
                             '()))         ; bound-vars
        (else (error "bad proc: " proc))))
```

Suppose we changed it so that we no longer called `substitute` inside of `apply-1`:

```
(define (apply-1 proc args)
  (cond ((procedure? proc)
         (apply proc args))
        ((lambda-exp? proc)
         (eval-1 (caddr proc)))  ; evaluate the body without using substitute
        (else (error "bad proc: " proc))))
```

With the above changes, show what Scheme-1 would print given the following inputs. If the result is an error, just write "error".

```
Scheme-1: ((lambda (x) (* x x)) 3)
ERROR

Scheme-1: ((lambda (x) (+ 2 3)) 2)
5

Scheme-1: (+ 3 4)
7

Scheme-1: (lambda (x) (* x x))
(lambda (x) (* x x))

Scheme-1: (lambda (x) (+ 3 4))
(lambda (x) (+ 3 4))

Scheme-1: ((lambda (cons) (cons 1 2)) +)
(1 . 2)
```

Exercise 11. Write a function `max-children` that returns the maximum number of children of any node in the tree. For example, suppose the following tree is `mytree`:



In this case, `(max-children mytree)` would return 3.

```
(define (max-children tree)
  (if (null? (children tree))
      0
      (max (length (children tree)) (max-children-forest (children tree)))))

(define (max-children-forest forest)
  (apply max (map max-children forest)))
```